
Application Note #2435 
 

Using the Ethernet to Communicate Between Multiple Controllers 
 
 
For multi-controller stand-alone applications, sometimes it is critical that there is a way 
of sending information between the controllers.  This enables the user to expand his/her 
control to many more axes and over much larger distances (if needed).  The ability to do 
this is due to the controller’s Ethernet communication channel that has 8 handles per 
controller.  Since a connection to another controller only uses up one handle, it is possible 
to connect 7 other controllers while still having an ethernet communication channel free 
to talk to the PC (or Graphical User Interface).  So, with each controller having the 
maximum of 8 axes, a 64 axis stand-alone system is entirely possible.  To keep it simple, 
we will limit the examples to just show how to communicate between  two multi-axis 
controllers.   
 
This application note deals with standard ethernet controllers such as the DMC-2100, 
2200, 21x2, 21x3, 1415,1416, and 1425.  Standard straight-through “patch” cables should 
connect each controller to a hub.  For 1 and 2 axis “Distributed Controllers” such as the 
DMC-3415, 3425 – this application note is not necessary since they have special 
commands that enable multi-axis distributed control systems – see the DMC-3425 User 
Manual for more info on this. 
 
The first example shown here shows how the first controller can poll the status of the 
second controller.  In this case, controller #1 displays the message “MOVING” when it 
sees that the operand _BGX on controller #2 shows there is motion on the X axis.  It then 
displays “STOPPED” when the controller finishes the move.  To test this, assign 
controller #1 the IP address 100.0.151.2011 and controller #2 the IP address 
100.0.151.202.1  Then, load the code below onto controller #1.  Type in CFA if 
connected over ethernet and CFS if connected over the Serial port in order to get the 
unsolicited messages to come back to the right communication channel.  Then type in 
CW2 to configure proper communications and XQ to execute the program.  Next, send 
the commands “SP10000;IP10000” to the second controller and you should see the first 
controller report back that the axis is moving. 
 
#INIT 
IHD=>-2;'Close Handle D 
WT200 
IHD=100,0,151,202;'Open Handle D 
#L1 
JP#L1,_IHB2<>-2 
MG"CONNECTED TO 100.0.151.202" 
#LOOP 
SAD="MG_BGX" 
WT250 
V1=_SAD0 

MGV1 
JP#LOOP,V1=0 
MG"MOVING" 
#MOVING 
SAD="MG_BGX" 
WT250 
V1=_SAD0 
MGV1 
JP#MOVING,V1=1 
MG"STOPPED" 
JP#LOOP 
 
 
In the above example, there is no need to have any code on the 2nd controller, however 
there are times when it might make sense to have both controllers open a connection to 
each other.  In this next example, there is code that needs to be downloaded and run on 
each controller.  Start up a SmartTERM window for each controller over an Ethernet 
connection.  Type in TH (Tell Handle) to make sure that the only connection to each 
controller is from the PC on Handle A.  Next type in CFA;CW2 to both controllers to 
make sure the Terminal is able to see unsolicited messages come back from the 
controllers.  Then download the following code to the appropriate controller and type in 
XQ on Controller #1. 
 
Controller #1 Code 
#INIT 
CFA;'Send Messages to PC thru handle A 
IHD=>-2;'Close Handle D if open 
WT200 
IHD=100,0,151,202;'Open Handle D 
#L1 
JP#L1,_IHD2<>-2 
MG"CONNECTED TO 100.0.151.202" 
SAD="XQ#INIT";'Initialize Slave 
WT2000 
#RUN 
SpeedX=5000 
PosX=10000 
#MOVE 
SAD="SPX=",SpeedX 
SAD="PRX=",PosX 
SAD="BGX" 
MG"Moving Slave X-axis" 
SAD="XQ#XDONE" 
#L2 
WT200 
SAD="RPX" 
WT200 
MG"Slave X= "{N};MG_SAD0 
JP#L2,XDONE=1 
MG"WAITING FOR NEW SpeedX & PosX" 
WT2000 
JP#MOVE 
EN 

Controller #2 Code: 
#INIT 
CFA 
IHC=>-2;'Close Handle C 
WT200 
IHC=100,0,151,201;'Open Handle C 
#L1 
JP#L1,_IHC2=-2 
MG"CONNECTED TO 100.0.151.201" 
EN 
#XDONE 
XDONE=_BGX 
XDONE=? 
SAC="XDONE=",XDONE 
WT100 
JP#XDONE,XDONE=1 
MG"FINISHED" 
EN 
 
 
 
There are a few different ways of getting the controllers to talk to each other, however the 
“SA” command is certainly the preferred method.  An alternative method is to use MG 
commands that are directed out to the appropriate handle connection.  For instance, if 
controller #1 opens a connection on handle D to the second controller, and then the 
command MG “V1=123”{ED} is given, it will send the string “V1=123” out the D 
handle to the second controller.  When sending variables, or commands where a response 
is not needed, this is a valid method of communication, however this will not work to 
query a value on the 2nd controller.  For instance, you cannot send a “TPX” command 
with the MG method since there is no way to get the response back.  
 
It is also possible for the “slave” controller to communicate back to the “master” 
controller on the same handle.  In other words, there only technically needs to have one 
handle connection opened.  Once the IH command is given, and a valid connection is 
established, both controllers can send and receive packets on the same channel.  Be 
careful to keep track of which handles are opened/closed on each controller as this could 
cause a lot of headaches if left unnoticed.  Some commands that are helpful when dealing 
with Ethernet communication are:  IA, IH, TH, WH, CF, MG, CW2, #TCPERR. 
 
 
 
Tips & Tricks 
 
1 It is critical with Ethernet communication between controllers to be on a “local” 
network that is not connected to the “company” network.  IP address assignment is 
dependant on the IP address of the user’s PC.  To check a PC’s IP address in Windows, 
open a DOS shell and type in: ipconfig.  The IP address that is assigned to the controllers 
should closely resemble the IP address of your PC (ie: in this example the PC’s address is 
100.0.151.10). 


